Evaluating EV1-techniques for estimating upper quantiles of TCEV data

Martin A.J. van Montfort[1, 2, 3, 4]

Contents

Summary

1. Introduction

2. Multi-annual TCEV maxima and EV1

3. Inference on quantiles of annual maxima using bi-annual maxima and EV1-techniques

3.1 A simple procedure, some competitors and their quality for quantile estimation

3.2 Evaluation of the estimators by simulation

3.3 Concluding remarks

References

Appendices

App. 1 Efficiency of estimating the *p*-point of the annual EV1-maximum from maxima over non-overlapping *k*-annual periods

App. 2 Simulation program

---

[1] Dept. of Mathematics (Statistics division), Wageningen Agricultural University; Dreijenlaan 4; 6703 HA Wageningen; The Netherlands. (montfort@rcl.wau.nl)

[2] Report of a research activity during a sabbatical stay (July/Aug. 1996) at NIWA (National Institute for Water and Atmosphere), Wellington, New Zealand.

[3] Handout at 7th International Meeting on Statistical Climatology, Whistler BC Canada (May 25-29, 1998).

[4] Dedicated to K.John A. Revfeim (1938-1998).

Evaluating EV1-techniques for estimating upper tail quantiles of TCEV data

Martin A.J. van Montfort

Summary

This paper deals with the quality of estimators of quantiles in the upper tail of TCEV-distributed annual maxima where the estimation technique is based on an EV1-approach of bi-annual maxima. The relative root mean square error of the quantile estimator is quantified by simulation.

1. Introduction

Modelling annual maxima of e.g. climatological events could start with a multi-component approach; for instance: split the year in (M =) 12 months or 4 seasons each with its own maximum with cdf C_m (m = 1, ..., M) and assume independence of the M maxima. Then the cumulative distribution function (cdf) of the annual maximum, denoted by A_1, is equal to the product $\prod_{m=1}^{M} C_m$.

Here we pay attention to CPE (Compound Poisson Exponential distribution of the maximum of a Poisson count of events of Exponential size) with support [0, ∞), which is close to the EV1 (Extreme Value distribution, type 1) with support (−∞, +∞). The difference is that the probability for negative values of the EV1 is the point probability on zero of the CPE.

The annual maximum could be modelled by a TCEV (Two Components Extreme Value distribution) in case of two seasons as a special case of the MCEV (Multi CEV) in case of a split in M subperiods.

A different argument for TCEV could be used. Revfeim (1991) modelled the M subperiod maxima with M CPE's, each with a Poisson rate parameter (ρ_m) and an Exponential size parameter (μ_m), which model contains 2M parameters. He reduced the number of parameters by series expansion to four ($\bar{\rho}$, $\bar{\mu}$, $C_{\mu\mu}$, $C_{\rho\mu}$), and his final distribution is close to the TCEV distribution.

Revfeim's 4-parameter distribution for $M \geq 2$ reads

$$-\ln(A(x)) = M\bar{\rho} \cdot \exp(-\tilde{x})\{1 + C_{\rho\mu}\tilde{x} + C_{\mu\mu}(\tilde{x}^2/2 - \tilde{x})\}$$

with in general	and for M = 2:
$\bar{\rho} = \Sigma\rho_m/M$	$\bar{\rho} = (\rho_1 + \rho_2)/2$
$\bar{\mu} = \Sigma\mu_m/M$	$\bar{\mu} = (\mu_1 + \mu_2)/2$
$\tilde{x} = x/\bar{\mu}$	
$C_{\mu\mu} = \Sigma(\mu_m/\bar{\mu} - 1)^2/M$	$C_{\mu\mu} = (\mu_1 - \mu_2)^2/(\mu_1 + \mu_2)^2$

$$C_{\rho\mu} = \Sigma\{(\rho_m/\bar{\rho} - 1)(\mu_m/\bar{\mu} - 1)\}/M \qquad\qquad C_{\rho\mu} = \frac{(\rho_1 - \rho_2)(\mu_1 - \mu_2)}{(\rho_1 + \rho_2)(\mu_1 + \mu_2)}$$

This formula represents a CPE for $\mu_1 = ... = \mu_M$ ($\rightarrow C_{\mu\mu} = C_{\rho\mu} = 0$), which is in agreement with the theory; the restriction leads to the formula with $M = 1$.
For a TCEV distribution one has

$$-\ln(cdf) = \rho_1 e^{-x/\mu_1} + \rho_2 e^{-x/\mu_2}.$$

Given $\bar{\rho}$, $\bar{\mu}$, $C_{\mu\mu}$, $C_{\rho\mu}$ for $M \geq 2$ one could choose

$$\mu_1 = \bar{\mu}(1 - \sqrt{C_{\mu\mu}})$$
$$\mu_2 = \bar{\mu}(1 + \sqrt{C_{\mu\mu}})$$
$$\rho_1 = \bar{\rho}(1 - C_{\rho\mu}/\sqrt{C_{\mu\mu}})$$
$$\rho_2 = \bar{\rho}(1 + C_{\rho\mu}/\sqrt{C_{\mu\mu}})$$

to be parameters of a TCEV which will be in close agreement with Revfeim's 4 parameter distribution. Van Montfort (1989) presented ML-estimation for Revfeim's 4-parameter distribution.

This introduction could be considered as support for using TCEV for annual maxima including the statistical tool kit for estimating the parameters and quantiles (e.g. 10- and 100-years return values) for TCEV-data; so this toolkit has to contain more than only EV1-techniques. If only EV1-techniques are available, then applying these to TCEV-data would introduce bias and would higher the mean square error (MSE).

An idea to reduce bias as a result of an inadequate model is used by Stephen Thompson, see Jowett and Thompson (1977). It has two aspects:

(i) the maximum of two independent and identically distributed (iid) TCEV (ρ_1, μ_1; ρ_2, μ_2)-variates is a TCEV($2\rho_1$, μ_1; $2\rho_2$, μ_2)-variate, and

(ii) the last mentioned variate is in some sense more close to the EV1 than the first one (this has to be formulated more carefully and has to be proved; see section 2).

This idea results in the procedure:
analyse non overlapping bi-annual maxima with cdf A_2 by EV1-techniques for inference on TCEV annual maxima, noting that the p-point, x_p, of the annual maximum equals $A_2^{-1}(p^2)$ based on $p = A_1(x_p)$ and $A_2(x) = A_1^2(x)$. For a graphical representation, see Fig. 1.

This paper supports the statement that the bi-annual maximum is more close to EV1 in the uppertail. It also compares some EV1- and other techniques applied to TCEV-data with respect to the relative root MSE for estimating quantiles in the right tail (10- and 100-years return values).



Figure 1. EV1-plot of CPE(ρ_I,μ_{II}) and CPE(ρ_{II},μ_{II}), of the annual maximum A_1 ~ TCEV(ρ_I,μ_I; ρ_{II},μ_{II}) and of the k-annual maximum A_k ~ TCEV($k\rho_I$,μ_I; $k\rho_{II}$,μ_{II}).

2. Multi-annual TCEV maxima and EV1

The point of interest is the behaviour of the k-year maximum, being the maximum of k iid TCEV variates each with distribution TCEV (ρ_1, μ_1; ρ_2, μ_2). The distributions of the independent components I and II are CPE (ρ_1, μ_1) and CPE(ρ_2, μ_2). So we get TCEV ($k\rho_1$, μ_1; $k\rho_2$, μ_2) for the k-year maximum. For convenience we assume $\mu_1 < \mu_2$; for $\mu_1 = \mu_2$ (= μ) we get CPE($k(\rho_1+\rho_2)$, μ).

Two arguments for the k-year maximum to get an EV1-behaviour are the following.

(i) The asymptotic ($k \rightarrow \infty$) behaviour of a maximum of k iid variates knows three types, depending on the right tail behaviour of the distribution of the individual maxima, with a TCEV distribution overhere; now the Exponential behaviour of the TCEV leads asymptotically to an EV1, and for k finite the EV1 could be a good approximation.

(ii) The second argument focusses on the two components of the k-annual maximum, being the maximum of two independent variates with cdf CPE($k\rho_1$, μ_1) and CPE($k\rho_2$, μ_2). Based on $\exp(-\rho e^{-x/\mu}) = \exp\{-\exp(-x - \mu\ln(\rho))/\mu\}$, connecting the CPE- and EV1-notation, we see that the mode $\mu\ln(\rho)$ is a location parameter and μ a scale parameter. Table 1 presents the location and scale parameters of the components of the k-annual maximum (m_k).

Table 1 Parameters of the maximum (m_k) of k iid TCEV variables.

	Component I		Component II	
variate	mode	scale	mode	scale
m_k	$\mu_1\ln(k\rho_1)$	μ_1	$\mu_2\ln(k\rho_2)$	μ_2
$(m_k - \text{mode}_{II})/\text{scale}_{II}$	$\dfrac{\mu_1\ln(k\rho_1) - \mu_2\ln(k\rho_2)}{\mu_2}$	$\dfrac{\mu_1}{\mu_2} < 1$	0	1

Analogously the parameters of the components are presented for the standardised version of m_k being

$$(m_k - \text{mode}_{II}) \,/\, \text{scale}_{II}.$$

This variate is the maximum of a contribution of II, being a standard EV1, and a contribution of I with scale parameter < 1 and location parameter:

$$\ln(k)\left(\frac{\mu_1}{\mu_2} - 1\right) + \left\{\frac{\mu_1}{\mu_2}\ln(\rho_1) - \ln(\rho_2)\right\}$$

The second term does not depend on k, and vanishes at equal modes of the components: $\mu_1\ln(\rho_1) = \mu_2\ln(\rho_2)$. The first term, with $\mu_1/\mu_2 - 1 < 0$ as a result of the assumption $\mu_1 < \mu_2$, is a negative multiple of $\ln(k)$.
So, the linearly rescaled m_k is the maximum of a standard EV1 and a non standard EV1 with scale parameter < 1 and a location parameter shifting from 0 (for $k = 1$) to $-\infty$ (for $k \to \infty$). Now it is (at least intuitively) clear that m_k for growing k tends to behave like an EV1-variate (with a decreasing anomaly in the left tail).

The vanishing influence of component I for increasing k can be quantified in the following way. Let X be the maximum of the independent maxima X_1 and X_2 with distribution CPE(ρ_1, μ_1) and CPE(ρ_2, μ_2), with cdf $F_1(x_1)$ and $F_2(x_2)$, and with density $f_1(x)$ and $f_2(x)$ where $f(x) = (\rho/\mu)e^{-x/\mu}\cdot F(x)$. Omitting $P(X_1 = 0) = \exp(-\rho_1)$ and $P(X_2 = 0) = \exp(-\rho_2)$, which is reasonable for ρ_1 and ρ_2 not to small, we get for $P\{X_I > X_{II}\}$, denoted by p_I:

$$P\{X_I > X_{II}\} = \int_{t=0}^{\infty} f_1(t)\cdot F_2(t)\,dt$$

$$= \int_{t=0}^{\infty} \frac{\rho_1}{\mu_1}\cdot\exp\{-[\frac{t}{\mu_1} + e^{-\rho_1 t/\mu_1} + e^{-\rho_2 t/\mu_2}]\}\,dt$$

which is a function of ρ_1, ρ_2 and μ_1/μ_2. The rescaling of m_k, applied above, supports that it is a function of μ_1/μ_2 and $\rho_1^{\mu_1/\mu_2}/\rho_2$.
With $x = \mathrm{arctg}(t) \cdot 2/\pi$, $t = \mathrm{tg}(x \cdot \pi/2)$, $dt/dx = (\pi/2)/\cos^2(x.\pi/2)$ we get

$$P\{X_{\mathrm{I}} > X_{\mathrm{II}}\} = \int_{x=0}^{1} \frac{\rho_1}{\mu_1} \cdot \exp\{-[...]\} \cdot \pi/2 / \cos^2(x \cdot \pi/2)\, dx$$

which is straightforward for numerical integration. Replacing ρ_1 and ρ_2 by $k\rho_1$ and $k\rho_2$ has to show a decreasing integral.

Beran et al. (1986) present $p_{\mathrm{II}} = P\{X_{\mathrm{I}} < X_{\mathrm{II}}\}$ as a function of $\theta = \mu_2/\mu_1$ and $\lambda = \rho_1/\rho_2^{1/\theta}$:

$$p_{\mathrm{II}} = \frac{\lambda}{\theta} \sum_{j=0}^{\infty} t_j$$

with

$$t_0 = \Gamma(1/\theta)$$
$$t_j = t_{j-1} \cdot \frac{-\lambda}{j} \frac{\Gamma((j+1)/\theta)}{\Gamma(j/\theta)}.$$

This formula reduces to $p_{\mathrm{II}} = \rho_2/(\rho_1 + \rho_2)$ for $\mu_1 = \mu_2$. Their Fig. 3 presents p_{II} as a function of θ and $\theta\ln(\lambda)$.

3. Inference on quantiles of annual maxima using bi-annual maxima and EV1-techniques.

3.1 A simple procedure, some competitors and their quality for quantile estimation

The proposal of Stephen M. Thompson, see Jowett & Thompson (1977), for inference on annual maxima is the following. Collect bi−annual maxima over non−overlapping time intervals, fit the EV1, assume independence in the series of annual maxima and use the estimated p^2−point of the bi−annual maximum as an estimate of the p−point of the annual maximum, being the $1/(1-p)$−years return value till exceedance. In the sequel this procedure is called the SMT−procedure. This proposal aims to reduce deviations of the EV1−model in the left tail of annual maxima, but halves the number of data. A more general proposal could be to use k−annual maxima and their estimated p^k−point.

The efficiency of this method at EV1−annual maxima, see Appendix 1, for estimating 10−years return value ($x_{0,90}$) from bi−annual maxima is 0.79, and for the 100−years return value 0.66.

The estimation technique could be e.g. Probability Weighted Moments, see Hosking (1985), or Maximum Likelihood. Here we use PWM.

The SMT−procedure invites to study some also simple competitors.

(i) Use all n annual maxima and fit EV1−parameters.

(ii) The restriction of non−overlapping bi−annual maxima reduces the number of independent observations to $n/2$ at n annual maxima. Using all $n(n-1)/2$ pairs irrespective of the chronological ordering introduces dependence; neglecting this dependence leads to a simple estimator of EV1−parameters.

(iii) Fit EV1−parameters to annual maxima after left censoring of $n/4$ observations. Cunnane (1987, pp. 60-61) has pointed out that "Censoring from below as a matter of course might be advantageous. It may be that smaller sample values have only nuisance value in the context of upper quantile estimation and also in model form testing and verification".

(iv) Fit GEV−parameters to all annual maxima.

As a measure of lack of quality of a procedure for estimating x_p we use the expected squared relative error

$$E\{(\hat{x}_p - x_p)/x_p\}^2 = \{\mathrm{var}(\hat{x}_p) + \mathrm{bias}^2\}/x_p^2 = (\mathrm{bias}/x_p)^2 + \mathrm{cv}^2$$

or its square root, being the relative root mean square error (rRMSE). This measure contains two components: the relative bias and the coefficient of variation of $\hat{x}_p$.
At complete samples PWM−estimators were used, see Hosking et al. (1985). In case of censoring type II (ranks of the observations to be collected are known before getting the data) with at the left skipping a fraction p_c of the ordered observations $x_{(1)},\ldots,x_{(n)}$, in that case the estimation technique for type I censoring (interval in which observations are collected fixed in advance) was applied using data with ranks exceeding np_c and with size larger than the estimated lower bound of the interval being the observation with rank $[np_c]$. For estimation at type I data, see Van Montfort (1997) and Cohen (1965).

3.2 Evaluation of the estimators by simulation

Here the problem is to study small sample qualities of estimators of the p−quantile, x_p, where the estimators used are meant to have good qualities for a different type of distribution, so in case of a slightly inadequate model.
An algebraic approach of the MSE, with components squared bias and variance, seems to be too cumbersome. The lack of quality measure is the ratio of the root MSE($\hat{x}_p$) and x_p, so the relative root MSE (rRMSE). This measure has to depend on six items: n (sample size), p and the four TCEV−parameters $\rho_1,\mu_1,\rho_2,\mu_2$ where only the ratio of μ_2 and μ_1 is essential. Practical relevance supported the following choices: n = 20, 32, 50; p = 0.90, 0.99 (associated with the 10− and 100−years return value). Still 3 choices are left. A restriction could be to fix the ratio (R) of the modes of the second and first component: $\mu_2 \ln(\rho_2) = R\mu_1\ln(\rho_1)$; here we deal with equal modes (R = 1).
Here we study a TCEV−system with second component CPE (ρ_2 = 4, μ_2 = 24). The deviation of the TCEV compared to the EV1 at equal modes could be quantified by μ_2/μ_1

(see section 2). A different quantity could be the difference on a logit scale of p_2 (probability that the maximum comes from the second component, so from the component with the larger μ) between TCEV($2\rho_1,\mu_1;2\rho_2,\mu_2$) and TCEV($\rho_1,\mu_1;\rho_2,\mu_2$).
Table 1 presents the choices made (apart from ρ_2 = 4, μ_2 = 24 resulting in mode = 33.271).

Table 1 Input of simulation study

ρ_1	μ_1	μ_2/μ_1	$(p_2)_{ann}$	$(p_2)_{bi}$	diff. on logit scale	$x_{0.90}$	$x_{0.99}$
8	16	1.5	0.52	0.60	0.28	93.3	145.9
16	12	2	0.55	0.66	0.46	89.5	143.9
32	9.6	2.5	0.56	0.70	0.59	88.0	143.7
256	6	4	0.58	0.76	0.83	87.3	143.7
1024	4.8	5	0.59	0.78	0.92	87.3	143.7
4096	4	6	0.60	0.80	0.98	87.3	143.7

The quality of estimators was quantified by generating 1000 pseudo random samples, so utilising the idea of blocking by comparing estimators applied to the same sample. For the results, see Table 2. The simulation program is added as Appendix 2.

Table 2 Simulation results.

ρ_1	μ_1	ρ_2	μ_2	n	rRMSE $*\sqrt{n}$	EV1	bian. all	0.25-censor	GEV
Return period: 10									
8.0	16.0	4.0	24.0	20	0.6453	0.886	0.953	0.881	0.863
8.0	16.0	4.0	24.0	32	0.6471	0.895	0.975	0.884	0.880
8.0	16.0	4.0	24.0	50	0.6583	0.887	0.975	0.865	0.869
16.0	12.0	4.0	24.0	20	0.6807	0.879	0.950	0.867	0.846
16.0	12.0	4.0	24.0	32	0.6873	0.883	0.967	0.863	0.855
16.0	12.0	4.0	24.0	50	0.6958	0.878	0.973	0.852	0.849
32.0	9.6	4.0	24.0	20	0.7035	0.882	0.951	0.871	0.848
32.0	9.6	4.0	24.0	32	0.7128	0.884	0.965	0.868	0.855
32.0	9.6	4.0	24.0	50	0.7205	0.883	0.972	0.864	0.854
256.0	6.0	4.0	24.0	20	0.7285	0.908	0.958	0.913	0.899
256.0	6.0	4.0	24.0	32	0.7388	0.916	0.967	0.924	0.920
256.0	6.0	4.0	24.0	50	0.7446	0.929	0.975	0.946	0.948
1024.0	4.8	4.0	24.0	20	0.7339	0.920	0.960	0.935	0.932
1024.0	4.8	4.0	24.0	32	0.7439	0.933	0.969	0.951	0.962
1024.0	4.8	4.0	24.0	50	0.7495	0.951	0.977	0.982	1.004
4096.0	4.0	4.0	24.0	20	0.7366	0.929	0.961	0.948	0.957
4096.0	4.0	4.0	24.0	32	0.7473	0.943	0.969	0.968	0.993
4096.0	4.0	4.0	24.0	50	0.7524	0.966	0.977	1.003	1.046
Return period: 100									
8.0	16.0	4.0	24.0	20	0.8636	0.794	0.960	0.812	1.405
8.0	16.0	4.0	24.0	32	0.8610	0.807	0.982	0.826	1.407
8.0	16.0	4.0	24.0	50	0.8726	0.813	0.988	0.823	1.395
16.0	12.0	4.0	24.0	20	0.9029	0.832	0.978	0.841	1.403
16.0	12.0	4.0	24.0	32	0.9102	0.861	0.996	0.867	1.406
16.0	12.0	4.0	24.0	50	0.9235	0.904	1.008	0.910	1.395
32.0	9.6	4.0	24.0	20	0.9275	0.873	0.985	0.881	1.407
32.0	9.6	4.0	24.0	32	0.9402	0.918	1.001	0.923	1.414
32.0	9.6	4.0	24.0	50	0.9567	0.984	1.015	0.993	1.409
256.0	6.0	4.0	24.0	20	0.9468	0.941	0.988	0.951	1.452
256.0	6.0	4.0	24.0	32	0.9605	1.010	1.001	1.012	1.489
256.0	6.0	4.0	24.0	50	0.9765	1.108	1.017	1.115	1.533
1024.0	4.8	4.0	24.0	20	0.9463	0.963	0.987	0.972	1.481
1024.0	4.8	4.0	24.0	32	0.9588	1.039	1.001	1.037	1.539
1024.0	4.8	4.0	24.0	50	0.9737	1.145	1.015	1.145	1.612
4096.0	4.0	4.0	24.0	20	0.9450	0.975	0.985	0.983	1.504
4096.0	4.0	4.0	24.0	32	0.9560	1.057	0.999	1.050	1.579
4096.0	4.0	4.0	24.0	50	0.9702	1.167	1.014	1.160	1.674

The sample sizes n = 20, 32, 50 were chosen to be nearly equidistant in the square root scale. Column 6 of Table 2 shows that $r\text{RMSE}\cdot\sqrt{n} = \{n\cdot\text{MSE}(\hat{x}_p)/x_p^2\}^{1/2}$ does nearly not depend on n.

The columns 7,...,10 present rRMSE of a method as a multiple of rRMSE of the SMT−procedure on bi−annual non−overlapping chronological pairs. For values > 1 the method is inferior compared to the SMT−procedure.

The top halve of Table 2 shows results for estimating of the 10 years return value, being an interpolation at the sample sizes used; the bottom halve deals with extrapolation to get the 100−years return value.

Table 2, column 8, shows that the gain by using all bi−annual pairs, compared to only non−overlapping chronological bi−annual pairs, is moderate (perhaps the estimation procedure used is too far from optimal?).

Table 2, column 9, shows that left consoring does a reasonable job, especially for small μ_2/μ_1.

Table 2, column 10, shows that GEV, with 1 parameter more, does a good job for small μ_2/μ_1 in case of interpolation but fails at extrapolation.

Table 2, column 7, shows that approximating a TCEV by an EV1 works well except for large μ_2/μ_1 at extrapolation.

The SMT−technique seems to have its best performance at extrapolation, in case of very rare large events combined with a hudge number of small events.

3.3 Concluding remarks

The SMT−procedure turns out to do a reasonable job for extrapolation at μ_2/μ_1 large. This size ratio is strongly connected, see Table 1, with the p_2−difference for bi−annual and annual maxima. For $(p_2)_{\text{bi−annual}}$ large compared to $(p_2)_{\text{annual}}$ the SMT−procedure promotes to get nearly only EV1−data, where the EV1−technique works well.

To get an idea which method works well one has to have an idea about $(\rho_1,\mu_1;\rho_2,\mu_2)$; then looking at Fig. 3 of Beran et al. (1986) tells the story of the increase of p_2 when going from annual to bi−annual maxima.

This paper dealt with a choice of the parameters of the large component of the TCEV (ρ_2 = 4, μ_2 = 24). Such a choice has to be based on knowledge of the local climate.

References

Beran, M., J.R.M. Hosking and N. Arnell. Comment on "Two-component extreme value distribution for flood frequency analysis". Water Resources Research 22 (1986) 263-266.

Cohen, A.C.. Maximum likelihood estimation in the Weibull distribution based on complete and on censored samples. Technometrics, 7 (1965), 579-588.

Cunnane, C. Review of statistical models for flood frequency estimation. In: Hydrological frequency modelling, edited by V.P. Singh, pp. 49-95, D. Reidel, Norwell, Mass., 1987.

Hosking, J.R.M., J.R. Wallis and E.F. Wood. Estimation of the generalized extreme-value distribution by the method of probability-weighted moments. Technometrics 27 (1985), 251-261.

Jowett, I.G. and S.M. Thompson (1977). Flows and design floods. The Clyde power project - Environmental impact report on design and construction proposals. Ministry of Works and Development, Wellington, New Zealand.

Landwehr, J.M., N.C. Matalas and J.R. Wallis. Probability weighted moments compared with some traditional techniques in estimating Gumbelparameters and quantiles. Water Resources Research 15 (1979), 1055-1064.

Revfeim, K.J.A.. Annual maxima and totals of seasonally varying processes. Stochastic Hydrology and Hydraulics 5 (1991), 147-153.

Rossi, F., M. Fiorentino and P. Versace. Two-component extreme value distribution for flood frequency analysis. Water Resources Research 20 (1984), 847-856.

Van Montfort, M.A.J.. Extreme-value distribution of annual maxima: ML estimation of the HGCP-model. Technical Note 89-06 of Department of Mathematics (Statistics division), Wageningen Agricultural University, 1989 (37 p.).

Van Montfort, M.A.J.. Concomitants of the Hershfield factor (1997). To appear in Journal of Hydrology (11 p.).

Appendix 1. Efficiency of estimating the *p*-point of the annual EV1-maximum from maxima over non-overlapping *k*-annual periods.

Let the annual maximum, m_1, follow an EV1(λ,σ) distribution with cdf $\exp(-\exp(-(x-\lambda)/\sigma))$, so λ is the mode and σ is a scale parameter. Let subsequent annual maxima be independent, and let the *k*-annual maximum be denoted by m_k.
Then m_k is EV1($\lambda+\sigma\ln(k)$, σ). The *p*−point of m_1, denoted by x_p, equals $\lambda+z_p.\sigma$ with $z_p = -\ln(-\ln p)$. The asymptotic ML covariance matrix at (λ,σ) reads

$$\frac{6\sigma^2/\pi^2}{n}\begin{pmatrix} e_{11} & e_{12} \\ e_{12} & e_{22} \end{pmatrix}$$

with $e_{11} = \pi^2/6 + (1-\gamma)^2$, $e_{12} = 1-\gamma$, $e_{22} = 1$ with γ is Euler's constant.
So on annual maxima ($k = 1$) we get

$$\mathrm{var}_1(\hat{x}_p) = \frac{6\sigma^2/\pi^2}{n}\{e_{11} + z_p^2 e_{22} + 2z_p e_{12}\}$$

Expressing x_p in the EV1−parameters of the *k*−annual maximum gives

$$x_p = \Lambda + z_{p^k}\cdot\Sigma$$

with

$$\Lambda = \lambda + \sigma\ln(k)$$
$$\Sigma = \sigma$$
$$z_{p^k} = z_p - \ln(k)$$

results in

$$\mathrm{var}_k(\hat{x}_p) = \frac{6\sigma^2/\pi^2}{n/k}\{e_{11} + (z_p - \ln k)^2 e_{22} + 2(z_p - \ln k)e_{12}\}.$$

The efficiency $E_{1,k}$ of n annual maxima versus n/k k−annual maxima for estimating x_p reads

$$E_{1,k} = \frac{\mathrm{var}_k(\hat{x}_p)}{\mathrm{var}_1(\hat{x}_p)} = k\frac{e_{11} + (z_p - \ln k)^2\cdot e_{22} + 2(z_p - \ln k)\cdot e_{12}}{e_{11} + z_p^2\cdot e_{22} + 2z_p\cdot e_{12}} = \frac{1}{E_{k,1}}$$

Some values of $E_{k,1}$ are:

p	0.50	0.80	0.90	0.95	0.99	optimum	
z_p	0.367	1.500	2.250	2.970	4.600	p	E
$E_{2,1}$	0.686	0.846	0.790	0.736	0.659	0.75	0.853
$E_{3,1}$	0.434	0.766	0.711	0.635	0.526	0.80	0.766
$E_{4,1}$	0.283	0.691	0.666	0.580	0.452	0.84	0.703

```
c       Asymptotic ML-efficiency of estimating a quantile of the EV1
c       of the annual maximum,using maxima over non-overlapping periods
c       of k years.
c       MM;June 1997.

        dimension eff(3),reff(3)

        open(unit=6,file='ev1eff1k.res',status='new')

        gam=0.5772156649
        e11= 3.1415927**2/6.+(1.-gam)**2
        e12=1.-gam
        e22=1.
        write (6,52)
        write (6,51)
        write (6,52)

        do 1 ip=1,99
        p=0.01*float(ip)
        z=-log(-log(p))
        rp=1./(1.-p)

          do 2 k=2,4
          flk=float(k)
          alk=log(flk)
          zz=z-alk
          a2=e11+e22*z **2+2.*z *e12
          a1=e11+zz**2*e22+2.*zz*e12
          eff(k)=flk*a1/a2
2         reff(k)=1./eff(k)
1       write (6,50) z,p,rp,(eff(k),reff(k),k=2,4)

51      format('       z        p       RP',
     & '     Eff(1,2)  (2,1)       (1,3)    (3,1)       (1,4)    (4,1)')
52      format(1x)
50      format(3f7.2,3(3x,2f8.3))
        stop
        end
```

z	p	RP	Eff(1,2)	(2,1)	(1,3)	(3,1)	(1,4)	(4,1)
-1.53	0.01	1.01	3.404	0.294	6.805	0.147	10.959	0.091
-1.36	0.02	1.02	3.411	0.293	6.882	0.145	11.162	0.090
-1.25	0.03	1.03	3.398	0.294	6.896	0.145	11.237	0.089
-1.17	0.04	1.04	3.376	0.296	6.879	0.145	11.250	0.089
-1.10	0.05	1.05	3.348	0.299	6.842	0.146	11.223	0.089
-1.03	0.06	1.06	3.316	0.302	6.790	0.147	11.167	0.090
-0.98	0.07	1.08	3.280	0.305	6.728	0.149	11.088	0.090
-0.93	0.08	1.09	3.242	0.308	6.656	0.150	10.991	0.091
-0.88	0.09	1.10	3.201	0.312	6.576	0.152	10.878	0.092
-0.83	0.10	1.11	3.158	0.317	6.490	0.154	10.752	0.093
-0.79	0.11	1.12	3.114	0.321	6.399	0.156	10.614	0.094
-0.75	0.12	1.14	3.068	0.326	6.302	0.159	10.465	0.096
-0.71	0.13	1.15	3.021	0.331	6.200	0.161	10.307	0.097
-0.68	0.14	1.16	2.973	0.336	6.095	0.164	10.141	0.099
-0.64	0.15	1.18	2.924	0.342	5.987	0.167	9.968	0.100
-0.61	0.16	1.19	2.875	0.348	5.876	0.170	9.789	0.102
-0.57	0.17	1.20	2.825	0.354	5.762	0.174	9.604	0.104
-0.54	0.18	1.22	2.774	0.360	5.646	0.177	9.414	0.106
-0.51	0.19	1.23	2.723	0.367	5.529	0.181	9.220	0.108
-0.48	0.20	1.25	2.673	0.374	5.410	0.185	9.023	0.111
-0.45	0.21	1.27	2.622	0.381	5.290	0.189	8.822	0.113
-0.41	0.22	1.28	2.571	0.389	5.169	0.193	8.620	0.116
-0.39	0.23	1.30	2.520	0.397	5.048	0.198	8.415	0.119
-0.36	0.24	1.32	2.470	0.405	4.927	0.203	8.209	0.122
-0.33	0.25	1.33	2.420	0.413	4.806	0.208	8.003	0.125
-0.30	0.26	1.35	2.370	0.422	4.685	0.213	7.796	0.128
-0.27	0.27	1.37	2.321	0.431	4.565	0.219	7.589	0.132
-0.24	0.28	1.39	2.273	0.440	4.445	0.225	7.382	0.135
-0.21	0.29	1.41	2.225	0.449	4.327	0.231	7.177	0.139
-0.19	0.30	1.43	2.178	0.459	4.210	0.238	6.973	0.143
-0.16	0.31	1.45	2.132	0.469	4.094	0.244	6.770	0.148
-0.13	0.32	1.47	2.087	0.479	3.979	0.251	6.569	0.152
-0.10	0.33	1.49	2.043	0.490	3.867	0.259	6.371	0.157
-0.08	0.34	1.52	1.999	0.500	3.756	0.266	6.175	0.162
-0.05	0.35	1.54	1.957	0.511	3.647	0.274	5.982	0.167
-0.02	0.36	1.56	1.916	0.522	3.540	0.282	5.792	0.173
0.01	0.37	1.59	1.876	0.533	3.435	0.291	5.605	0.178
0.03	0.38	1.61	1.837	0.544	3.333	0.300	5.421	0.184
0.06	0.39	1.64	1.799	0.556	3.233	0.309	5.241	0.191
0.09	0.40	1.67	1.762	0.568	3.135	0.319	5.065	0.197
0.11	0.41	1.69	1.726	0.579	3.040	0.329	4.892	0.204
0.14	0.42	1.72	1.692	0.591	2.948	0.339	4.724	0.212
0.17	0.43	1.75	1.658	0.603	2.858	0.350	4.560	0.219
0.20	0.44	1.79	1.626	0.615	2.770	0.361	4.400	0.227
0.23	0.45	1.82	1.595	0.627	2.686	0.372	4.244	0.236
0.25	0.46	1.85	1.566	0.639	2.604	0.384	4.092	0.244
0.28	0.47	1.89	1.537	0.651	2.524	0.396	3.945	0.253
0.31	0.48	1.92	1.510	0.662	2.448	0.409	3.802	0.263
0.34	0.49	1.96	1.484	0.674	2.374	0.421	3.664	0.273
0.37	0.50	2.00	1.459	0.686	2.302	0.434	3.530	0.283
0.40	0.51	2.04	1.435	0.697	2.234	0.448	3.400	0.294
0.42	0.52	2.08	1.412	0.708	2.168	0.461	3.275	0.305
0.45	0.53	2.13	1.391	0.719	2.105	0.475	3.155	0.317

0.48	0.54	2.17	1.370	0.730	2.045	0.489	3.039	0.329
0.51	0.55	2.22	1.351	0.740	1.987	0.503	2.927	0.342
0.55	0.56	2.27	1.333	0.750	1.931	0.518	2.820	0.355
0.58	0.57	2.33	1.316	0.760	1.879	0.532	2.717	0.368
0.61	0.58	2.38	1.300	0.769	1.829	0.547	2.619	0.382
0.64	0.59	2.44	1.285	0.778	1.781	0.561	2.524	0.396
0.67	0.60	2.50	1.271	0.787	1.736	0.576	2.434	0.411
0.70	0.61	2.56	1.258	0.795	1.693	0.591	2.348	0.426
0.74	0.62	2.63	1.246	0.803	1.653	0.605	2.267	0.441
0.77	0.63	2.70	1.235	0.810	1.615	0.619	2.189	0.457
0.81	0.64	2.78	1.224	0.817	1.580	0.633	2.115	0.473
0.84	0.65	2.86	1.215	0.823	1.546	0.647	2.046	0.489
0.88	0.66	2.94	1.207	0.828	1.515	0.660	1.980	0.505
0.92	0.67	3.03	1.200	0.833	1.487	0.673	1.918	0.521
0.95	0.68	3.13	1.193	0.838	1.460	0.685	1.860	0.538
0.99	0.69	3.23	1.188	0.842	1.436	0.696	1.806	0.554
1.03	0.70	3.33	1.183	0.845	1.414	0.707	1.756	0.570
1.07	0.71	3.45	1.179	0.848	1.394	0.718	1.709	0.585
1.11	0.72	3.57	1.176	0.850	1.376	0.727	1.666	0.600
1.16	0.73	3.70	1.174	0.852	1.360	0.735	1.626	0.615
1.20	0.74	3.85	1.173	0.853	1.346	0.743	1.590	0.629
1.25	0.75	4.00	1.172	0.853	1.334	0.750	1.557	0.642
1.29	0.76	4.17	1.173	0.853	1.324	0.755	1.528	0.654
1.34	0.77	4.35	1.174	0.852	1.317	0.760	1.503	0.665
1.39	0.78	4.55	1.176	0.851	1.311	0.763	1.481	0.675
1.45	0.79	4.76	1.178	0.849	1.307	0.765	1.462	0.684
1.50	0.80	5.00	1.182	0.846	1.305	0.766	1.447	0.691
1.56	0.81	5.26	1.186	0.843	1.305	0.766	1.436	0.697
1.62	0.82	5.56	1.191	0.839	1.308	0.765	1.428	0.701
1.68	0.83	5.88	1.197	0.835	1.312	0.762	1.423	0.703
1.75	0.84	6.25	1.204	0.830	1.319	0.758	1.422	0.703
1.82	0.85	6.67	1.212	0.825	1.327	0.753	1.425	0.702
1.89	0.86	7.14	1.221	0.819	1.338	0.747	1.432	0.698
1.97	0.87	7.69	1.230	0.813	1.352	0.740	1.443	0.693
2.06	0.88	8.33	1.241	0.806	1.367	0.731	1.458	0.686
2.15	0.89	9.09	1.253	0.798	1.386	0.722	1.477	0.677
2.25	0.90	10.00	1.266	0.790	1.407	0.711	1.502	0.666
2.36	0.91	11.11	1.281	0.781	1.432	0.698	1.532	0.653
2.48	0.92	12.50	1.297	0.771	1.461	0.685	1.568	0.638
2.62	0.93	14.29	1.315	0.761	1.493	0.670	1.611	0.621
2.78	0.94	16.67	1.335	0.749	1.531	0.653	1.663	0.601
2.97	0.95	20.00	1.358	0.736	1.575	0.635	1.724	0.580
3.20	0.96	25.00	1.385	0.722	1.628	0.614	1.800	0.556
3.49	0.97	33.33	1.417	0.706	1.693	0.591	1.894	0.528
3.90	0.98	50.00	1.458	0.686	1.777	0.563	2.021	0.495
4.60	0.99	100.00	1.518	0.659	1.903	0.526	2.213	0.452

Appendix 2. Simulation program.

```
c     Simulation to evaluate a proposal of S.M.Thompson (NIWA) to use
c     EV1-techniques on biannual maxima for inference on quantiles of
c     the annual maximum.

c     Here the annual maximum is modeled by TCEV,and the bi-annual maximum
c     could be approx. EV1.

c     methods to be compared:                     sample size    estimator

c     (1)  annual maxima EV1                      n                 PWM
c     (11) annual maxima GEV                      n                 PWM

c     (21) bi-annual maxima (no overlap)          n/2               PWM
c     (22) bi-annual maxima (all pairs )          n(n-1)/2          PWM

c     (3)  sorted annual maxima,skip left tail
c          ( e.g. 25 % )                          0.75*n
c  isw=1:  plotting position of x(i) based on
c          median of uniform(i),where uniform(i)
c          is replaced by beta(i,n+1-i);ordinary
c          least sqares (so:equal weights).Note
c          that this fitting procedure is far
c          from optimal.                                            OLS
c  isw=2   25 % censored at the left;approximated
c          lower bound equal to the largest
c          censored datum                                   (CPEXP  ML )

c     MM ( at NIWA;July,1996)
c        ( continued : mid 1997 )

c     IN  n1     sample size (has to be an even integer:2,4,...50);
c                end:n1=0.
c         nsim   number of simulations
c         ipr    =1 for extensive output
c         frskip fraction to be skipped in the left tail
c         prp    prob. related to m-years return value (p=1-1/m)
c         r1,s1  rate and size of one component of the TCEV
c         r2,s2  idem,other component

c     The mis-fit is based on  -1+the ratio of the estimated return value
c     and the return value in the model,from which the data come.

c     Calculated: the average mis-fit (bias)
c                 the average squared misfit ( rel MSE )
c                 its square root.

c     needed: IMSL: RNSET,RNUN,BETIN,GAMMA,SVRGN
c             local:TCEVINV,GEVPWM&SUBP3,EV1PWM,FIT,CPEXPML,PROB

      dimension d(50),d2(50),dd(1225),ru1(50),ru2(50),pp(50),dcen(50)

      open(unit=5,file='tcevsim.inp',status='old')
      open(unit=6,file='tcevsim.res',status='unknown')

6     read (5,50) n1,nsim,ipr,frskip,prp,r1,s1,r2,s2
      write (6,66)
      if ( n1 .eq. 0 ) goto 7

c       calculation of the probability that a TCEV maximum comes from
c       the second component ( incl. the difference at bi-annual vs.
c       annual maxima on logit scale );see: Beran et al. (1986).
        do 10 i=1,2
```

```
        r1l=r1*float(i)
        r2l=r2*float(i)
        call PROB(r1l,s1,r2l,s2,n2,ppp)
10      d(i)=ppp
        dif=log( d(2)*(1.-d(1)) /( (1.-d(2))*d(1) ))
        write(6,51) n1,nsim,frskip,prp,r1,s1,r2,s2,d(1),d(2),dif

      n1m1=n1-1
      n1c=n1*(n1-1)/2
      n2=n1/2
      n13=(float(n1)+0.5)*frskip

      r1l=log(r1)
      r2l=log(r2)
      zp=-log(-log(prp))
      zp2=zp-log(2.)
      call TCEVINV(r1,s1,r2,s2,prp,xp)

c     calculating plotting positions
         spp=0.0
         qpp=0.0
         pmed=0.5
         do 5 i=1,n1
         pin=i
         qin=n1+1-i
         bi=BETIN(pmed,pin,qin)
         ppp=-log(-log(bi))
         pp(i)=ppp
         if(i.lt.n13) goto 5
         spp=spp+ppp
         qpp=qpp+ppp**2
5        continue
         if(ipr .ne. 1) goto 8
         write (6,65)
         write (6,64) (pp(i),i=1,n1)
8        blun=n1+1-n13
         qpp=qpp-spp**2/blun

      q1 =0.0
      q11=0.0
      q21=0.0
      q22=0.0
      q3 =0.0

      ave1 =0.0
      ave11=0.0
      ave21=0.0
      ave22=0.0
      ave3 =0.0

c     ****************************************************************
c     SIMULATION: generating TCEV-data

      iseed=123457
      call RNSET(iseed)
      nc2=0
      do 1 k=1,nsim
         call RNUN(n1,ru1)
         call RNUN(n1,ru2)
         do 2 i=1,n1
         z1=-log(-log(ru1(i)))
         z2=-log(-log(ru2(i)))
         x1=s1*(z1+r1l)
         x2=s2*(z2+r2l)
         xx=x1
         if(x2 .le. x1) goto 9
```

```
         xx=x2
         nc2=nc2+1
9        d2(i)=xx
2        d(i)=xx

c     all single observations ( GEV- and EV1-fit)
         call GEVPWM(n1,d,gevp1,gevp2,gevp3,gevt3,p1,p2)
         xp1=p1+p2*zp
         rat=xp1/xp
         q1=q1+(rat-1.)**2
         ave1=ave1+rat
            xp11=gevp1+gevp2*(1.-exp(-gevp3*zp))/gevp3
            rat=xp11/xp
            q11=q11+(rat-1.)**2
            ave11=ave11+rat

c     censoring at the left side ( two estimation procedures :
c                                     isw=1 : OLS; isw=2 : ML.
         isw=2
         goto (11,12),isw

c        fitting straight line after skipping the left tail
11       call FIT(n13,n1,pp,qpp,spp,d,p1,p2)
         goto 13

c        ML-estimation
12       m=n13
         bound=d(m)
         mm=m+1
         nrncen=n1-m
         do 14 ii=1,nrncen
         iii=ii+m
14       dcen(ii)=d(iii)-bound
         p1=p1-bound
         rate=exp(p1/p2)
         size=p2
         if(ipr .ne. 1) goto 17
         write (6,50) m,n1
         write (6,15) (d(ii),ii=1,n1)
         write (6,15) (dcen(ii),ii=1,nrncen)
         write (6,15) p1,p2,rate,size
17       imax=20
         call CPEXPML(m,nrncen,dcen,rate,size,imax,it)
         p2=size
         p1=p2*log(rate)+bound

13       xp3=p1+p2*zp
         rat=xp3/xp
         q3=q3+(rat-1.)**2
         ave3=ave3+rat

c     maximum of all pairs
         li=0
         do 3 i1=1,n1m1
         i11=i1+1
         do 3 i2=i11,n1
         li=li+1
         xx=d(i1)
         if(d(i2) .gt. xx ) xx=d(i2)
3        dd(li)=xx
         call EV1PWM(n1c,dd,p1,p2)
c        r=exp(p1/p2)*0.5
c        p1=p2*log(r)
c        xp22=p1+p2*zp
         xp22=p1+p2*zp2
         rat=xp22/xp
```

```
         q22=q22+(rat-1.)**2
         ave22=ave22+rat

c     maximum of subsequent,non overlapping pairs
         do 4 i=1,n2
         i1=2*i-1
         i2=i1+1
         xx=d2(i1)
         if(d2(i2) .gt. xx ) xx=d2(i2)
4        d(i)=xx
         call EV1PWM(n2,d,p1,p2)
c        r=exp(p1/p2)*0.5
c        p1=p2*log(r)
c        xp21=p1+p2*zp
         xp21=p1+p2*zp2
         rat=xp21/xp
         q21=q21+(rat-1.)**2
         ave21=ave21+rat
      if(ipr.eq.1) write (6,60) k,xp, xp1,xp21,xp22,xp3, q1,q21,q22,q3
1     continue
c     ****************************************************************

      fl=float(nsim)
      frc2=float(nc2)/(float(n1)*fl)
      write (6,67) n13,n1,frc2
      q1=q1  /fl
      q11=q11/fl
      q21=q21/fl
      q22=q22/fl
      q3 =q3 /fl
      ave1 =ave1 /fl-1.
      ave11=ave11/fl-1.
      ave21=ave21/fl-1.
      ave22=ave22/fl-1.
      ave3 =ave3 /fl-1.
      sq1 =sqrt( q1)
      sq11=sqrt(q11)
      sq21=sqrt(q21)
      sq22=sqrt(q22)
      sq3 =sqrt( q3)
      write (6,61)
      write (6,63)              ave1,ave21,ave22,ave3,ave11
      write (6,62)  r1,s1,r2,s2,  xp,prp, q1,q21,q22,q3,q11
      write (6,63)                   sq1,sq21,sq22,sq3,sq11
      goto 6
7     stop

15    format(1x,10f7.3)
50    format(3i5,f10.5,f6.2,4f6.1)
51    format(2i5,2f6.2,4f7.1,3f6.3)
60    format(i5,5f8.2,4f8.4)
63    format(37x,5f8.4)
61    format(1x)
62    format(1x,5f6.1,f6.2,5f8.4)
64    format(1x,10f7.3)
65    format(' plotting positions')
66    format(' ************************************************'
     &       ,'**************************')
67    format(2i5,4x,' fraction data from 2nd component:',f8.3)

      end
```

```
c     *****************************************************************
      subroutine FIT(n13,n1,pp,qpp,spp,x,    p1,p2)
c                    in  in in in  in  in     out
c     ************************************************************  FIT

c     fitting a straight line (OLS) in an EV1-plot.
      dimension pp(50),x(50)
      sy=0.0
      syz=0.0
      do 1 i=n13,n1
      z=pp(i)
      y=x(i)
      sy=sy+y
1     syz=syz+y*z
      blun=n1+1-n13
      p2=(syz-sy*spp/blun)/qpp
      p1=(sy-p2*spp)/blun
      return
      end

c     *********************************************************
      subroutine CPEXPML(m,n,x,rate,size,imax,it)
c     ************************************************  CPEXPML
c                     IN + + +    +    +    +
c                    OUT + + +    +    +    +  +
c     input needs starting values for rate and size
c     Compound Poisson Exponential data: m point data at zero,
c     and n positive data in the array x.

      dimension x(1)

      it=0
      a=n
      rel=0.60
2     it=it+1
      if(it .gt. imax) return
      sz=0.0
      s0=0.0
      s1=0.0
      s2=0.0
      do 1 i=1,n
      z=x(i)/size
      ez=exp(-z)
      sz=sz+z
      s0=s0+ez
      s1=s1+ez*z
1     s2=s2+ez*z*z

      d10=-float(m)+a/rate-s0
      d20=-a/(rate*rate)

      d01=(-a+sz-rate*s1)/size
      d02=(a-2.*sz+2.*rate*s1-rate*s2)/(size*size)
      d11=-s1/size

      det=d20*d02-d11*d11
      c1=(-d02*d10+d11*d01)/det
      c2=( d11*d10-d20*d01)/det
      crit=(c1/rate)**2+(c2/size)**2
      rel=rel*rel
      f=1.-rel
      rate=rate+f*c1
      size=size+f*c2
      if(crit .lt. 1.E-07) return
      goto 2
      end
```

```
c     ********************************************************
      subroutine PROB(r1,s1,r2,s2,k,p)
c                     in in in in;out
c     **************************************************  PROB

c     TCEV with components CPE(r1,s1) and CPE(r2,s2) with rate
c     and size parameters r and s resp.
c       p is the probability that the TCEV-maximum comes from
c       the second component.
c       Reference:Beran.M.,J.R.M.Hosking and N.Arnell;
c       Comment on ...;Water Resources Research 22(1986)263-266.
c       formula (3) , to start with p=+ ( and not p=- )
c       input: r1,s1,r2,s2
c       output:p  (k is the number of terms used in the series)
c     IMSL is used for evaluating the gamma-function GAMM(.)

      crit=1.E-07
      kmax=100

      u =s2/s1
      rt=s1/s2
      v=r2*(r1**(-rt))
      t=GAMMA(rt)
      g1=t
      s=t
      k=0
1       k=k+1
        if ( k.eq.kmax) goto 2
        fk=k+1
        arg=fk*rt
        g2=GAMMA(arg)
        t=t*(-v)/(fk-1.)*g2/g1
        g1=g2
        s=s+t
        if ( abs(t*v*rt) .gt. crit ) goto 1
2     p=s*v*rt
      return
      end
```

```
c     ******************************************************************
      subroutine TCEVINV(r1,s1,r2,s2,p,x)
c     *******************************************************  TCEVINV

c     Inverse of the Two Component Extreme Value distribution:TCEV
c     Input: the two pairs of parameters for rate and size:(r1,s1),(r2,s2)
c            p is the value of the cdf.
c     Output: its p-point

c     Method used:Newton/Raphson,with EV1-approximated starting value.
c     ipr=1 for extensive output
      x=0.0
      if (p.le. exp(-(r1+r2))) return

      ipr=0
      imax=10
      iter=0
      relax=0.8
      z=-log(-log(p))
      x=(s1+s2)*0.5*(z+log(r1+r2))

1     iter=iter+1
      e1=exp(-x/s1)
      e2=exp(-x/s2)
      f0=log(p)+r1*e1+r2*e2
      f1=-(r1/s1*e1+r2/s2*e2)
      cor=-f0/f1
      relax=relax**2
      if (ipr .eq. 1) write (6,60) iter,r1,s1,r2,s2,p,x,cor
60    format(i4,4f5.1,f6.3,2f7.2)
      x=x+cor*(1.-relax)
      if( iter .eq. imax ) goto 2
      if( abs(cor/x) .gt. 0.00001 ) goto 1

2     return
      end

c     ***************************************************************
      subroutine GEVPWM(n,x, p1,p2,p3,t3,  gp1,gp2)
c                      in in;out
c     *****************************************************    GEVPWM

c     PWM-estimators for GEV- and EV1-parameters
c     TNMX 27(1985)251-261 ( Hosking et al.)

c     IN n   sample size
c        x   array of increasingly ordered data

c     OUT
c                 GEV  EV1
c     location     p1  gp1
c     scale        p2  gp2
c     shape        p3
c     lack of fit  t3       ( t3=p3/se)

      dimension x(1)

      a2=alog(2.)
      a3=alog(3.)
      fn=n
      f1=n-1
      f12=(n-1)*(n-2)
      b0=0.
      b1=0.
      b2=0.
      call SVRGN(n,x,x)
```

```
      do 1 j=1,n
      z=x(j)/fn
      g1=j-1
      g12=(j-1)*(j-2)
      b0=b0+z
      b1=b1+z*g1/f1
1     b2=b2+z*g12/f12
      cccc=(2.*b1-b0)/(3.*b2-b0)
      c=cccc-a2/a3
      p3=7.8590*c+2.9554*c*c
      cc=(3.*b2-b0)/(2.*b1-b0)
c     local iteration to get rid of the approximation
      call SUBP3(p3,cc,p3n)
      p3=p3n

      euler=0.57721566
      p3m1=p3-1.
      ga=GAMMA(p3m1)
      if(abs(p3).lt. 0.01) goto 2

      p2=(2.*b1-b0)*p3/(ga*(1.-2.**(-p3)))
      p1=b0+p2*(ga-1.)/p3
      goto 3

2     p2=(2.*b1-b0)/ga/(a2-p3*a2**2/2.+p3**2*a2**3/6.-p3**3*a2**4/24.)
      p1=b0+p2*(-euler+0.9891*p3-0.8574*p3*p3)
3      t3=p3*sqrt(fn/0.5633)

c     EV1-fit

      m=n/2
      b1=0.
      do 4 i=1,m
      ii=n+1-i
4     b1=b1+float(n+1-2*i)*(x(ii)-x(i))
      gp2=b1/(fn*f1*a2)
      gp1=b0-euler*gp2
      return
      end

c     *************************************************************
      subroutine SUBP3(p3,a,p3n)
c                     in in out
c     ******************************************************  SUBP3

      imax=10
      it=0
      p3n=p3
      a2=alog(2.)
      a3=alog(3.)
1     it=it+1
      d3=3.**(-p3n)
      d2=2.**(-p3n)
      f=d3-a*d2+a-1.
      fa=-a3*d3+a*a2*d2
      cor=-f/fa
      p3n=p3n+cor
      if(abs(cor) .lt. 0.0001 ) return
      if ( it .le. imax ) goto 1
      return
      end
```

```
c     ******************************************************************
      subroutine EV1PWM(n,x,    p1,p2)
c                      in in     out
c     **********************************************************  EV1PWM

c     PWM-estimators for EV1-parameters
c     see: TNMX 27(1985)251-261 ( Hosking et al.)

c     IN n   sample size
c        x   array of data ,to be ORDERED increasingly

c     OUT  location      p1
c          scale         p2

      dimension x(1)

      fn=n
      f1=n-1
      call SVRGN(n,x,x)

      b0=0.0
      do 1 j=1,n
1     b0=b0+x(j)
      b0=b0/fn
      m=n/2
      b1=0.0
      do 2 i=1,m
      ii=n+1-i
2     b1=b1+float(n+1-2*i)*(x(ii)-x(i))

      p2=b1/(fn*f1*alog(2.))
      p1=b0-0.57721566*p2

      return
      end
```